Problem:
List

For a dynamic list $L$ of integers, perform a sequence of the following operations. $L$ has a special element called END at the end of the list and an element of $L$ is indicated by a cursor.

insert($x$): Insert $x$ before the element indicated by the cursor. After this operation, the cursor points the inserted element.

move($d$): Move the cursor to the end by $d$, if $d$ is positive. Move the cursor to the front by $d$, if $d$ is negative.

erase(): Delete the element indicated by the cursor. After this operation, the cursor points the element next to the deleted element. In case there is no such element, the cursor should point END.

In the initial state, $L$ is empty and the cursor points END.

Input

The input is given in the following format.

$q$
$query_1$
$query_2$
:
$query_q$

Each query $query_i$ is given by

0 $x$

or

1 $d$

or

2

where the first digits 0, 1 and 2 represent insert, move and erase operations respectively.

Output

Print all elements of the list in order after performing given operations. Print an element in a line.

Constraints

$1 \leq q \leq 500,000$

The cursor indicates an element of $L$ or END during the operations

Erase operation will not given when the cursor points END

$-1,000,000,000 \leq x \leq 1,000,000,000$

Moving distance of the cursor ($\sum{|d|}$) does not exceed 1,000,000

$L$ is not empty after performing all operations

Sample Input 1

5
0 1
0 2
0 3
1 1
2

Sample Output 1

3
1

Solution in Haskell (judge verdict: Accepted)

import Control.Applicative
import Control.Monad
import Data.ByteString.Char8 (ByteString)
import qualified Data.ByteString.Char8 as B
import Data.List
import Data.Char (isSpace)

main :: IO ()
main = do
  q <- readLn
  solve <$> replicateM q f >>= mapM_ print
  where
    f = readil B.readInt <$> B.getLine

solve :: [[Int]] -> [Int]
solve = toListZ . foldl' f emptyZ
  where
    f z [0, x] = insertZ z x
    f z [1, d] = moveZ z d
    f z [2] = eraseZ z

readil :: Integral a =>  (ByteString -> Maybe (a, ByteString)) -> ByteString -> [a]
readil f = unfoldr $ \s -> do
  (n, s') <- f s
  return (n, B.dropWhile isSpace s')

type Zipper a = ([a], [a])

emptyZ :: Zipper Int
emptyZ = ([], [])

insertZ :: Zipper Int -> Int -> Zipper Int
insertZ (as, bs) x = (as, x:bs)

moveZ :: Zipper Int -> Int -> Zipper Int
moveZ z d = f z d
  where
    f (as, bs) d
      | d == 0 = (as, bs)
      | d > 0 = case bs of
                 (b:bs) -> f (b:as, bs) (d - 1)
                 [] -> (as, [])
      | otherwise = case as of
                     (a:as) -> f (as, a:bs) (d + 1)
                     [] -> ([], bs)

eraseZ :: Zipper Int -> Zipper Int
eraseZ (as, bs) = case bs of
                   [] -> (as, [])
                   (b:bs) -> (as, bs)

toListZ :: Zipper Int -> [Int]
toListZ (as, bs) = reverse as ++ bs